

## F07AWF (CGETRI/ZGETRI) – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F07AWF (CGETRI/ZGETRI) computes the inverse of a complex matrix  $A$ , where  $A$  has been factorized by F07ARF (CGETRF/ZGETRF).

### 2 Specification

```
SUBROUTINE F07AWF(N, A, LDA, IPIV, WORK, LWORK, INFO)
ENTRY      cgetri(N, A, LDA, IPIV, WORK, LWORK, INFO)
INTEGER    N, LDA, IPIV(*), LWORK, INFO
complex  A(LDA,*), WORK(LWORK)
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

To compute the inverse of a complex matrix  $A$ , the routine must be preceded by a call to F07ARF (CGETRF/ZGETRF), which computes the  $LU$  factorization of  $A$  as  $A = PLU$ . The inverse of  $A$  is computed by forming  $U^{-1}$  and then solving the equation  $XPL = U^{-1}$  for  $X$ .

### 4 References

- [1] Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

### 5 Parameters

- 1:** N — INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 2:** A(LDA,\*) — *complex* array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1,N)$ .  
*On entry:* the  $LU$  factorization of  $A$ , as returned by F07ARF (CGETRF/ZGETRF).  
*On exit:* the factorization is overwritten by the  $n$  by  $n$  matrix  $A^{-1}$ .
- 3:** LDA — INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F07AWF (CGETRI/ZGETRI) is called.  
*Constraint:*  $LDA \geq \max(1,N)$ .
- 4:** IPIV(\*) — INTEGER array *Input*  
**Note:** the dimension of the array IPIV must be at least  $\max(1,N)$ .  
*On entry:* the pivot indices, as returned by F07ARF (CGETRF/ZGETRF).
- 5:** WORK(LWORK) — *complex* array *Workspace*  
*On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.

**6:** LWORK — INTEGER *Input*

*On entry:* the dimension of the array WORK as declared in the (sub)program from which F07AWF (CGETRI/ZGETRI) is called.

*Suggested value:* for optimum performance LWORK should be at least  $N \times nb$ , where  $nb$  is the **blocksize**.

*Constraint:* LWORK  $\geq N$ .

**7:** INFO — INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ , the  $i$ th diagonal element of the factor  $U$  is zero,  $U$  is singular, and the inverse of  $A$  cannot be computed.

## 7 Accuracy

The computed inverse  $X$  satisfies a bound of the form:

$$|XA - I| \leq c(n)\epsilon|X|P|L||U|,$$

where  $c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the **machine precision**.

Note that a similar bound for  $|AX - I|$  cannot be guaranteed, although it is almost always satisfied. See Du Croz and Higham [1].

## 8 Further Comments

The total number of real floating-point operations is approximately  $\frac{16n^3}{3}$ .

The real analogue of this routine is F07AJF (SGETRI/DGETRI).

## 9 Example

To compute the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} -1.34 + 2.55i & 0.28 + 3.17i & -6.39 - 2.20i & 0.72 - 0.92i \\ -0.17 - 1.41i & 3.31 - 0.15i & -0.15 + 1.34i & 1.29 + 1.38i \\ -3.29 - 2.39i & -1.91 + 4.42i & -0.14 - 1.35i & 1.72 + 1.35i \\ 2.41 + 0.39i & -0.56 + 1.47i & -0.83 - 0.69i & -1.96 + 0.67i \end{pmatrix}.$$

Here  $A$  is nonsymmetric and must first be factorized by F07ARF (CGETRF/ZGETRF).

## 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07AWF Example Program Text
*      Mark 15 Release. MAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA, LWORK
PARAMETER       (NMAX=8,LDA=NMAX,LWORK=64*NMAX)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
*      .. Local Arrays ..
complex        A(LDA,NMAX), WORK(LWORK)
INTEGER          IPIV(NMAX)
CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL         cgetrf, cgetri, X04DBF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07AWF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
*      Read A from data file
*
READ (NIN,*) ((A(I,J),J=1,N),I=1,N)

*
*      Factorize A
*
CALL cgetrf(N,N,A,LDA,IPIV,INFO)

*
WRITE (NOUT,*)
IF (INFO.EQ.0) THEN

*
*      Compute inverse of A
*
CALL cgetri(N,A,LDA,IPIV,WORK,LWORK,INFO)

*
*      Print inverse
*
IFAIL = 0
CALL X04DBF('General', ' ', N,N,A,LDA, 'Bracketed', 'F7.4',
+          'Inverse', 'Integer', RLABS, 'Integer', CLABS, 80, 0,
+          IFAIL)
ELSE
WRITE (NOUT,*) 'The factor U is singular'
END IF
END IF
STOP
*
END

```

## 9.2 Program Data

F07AWF Example Program Data

```
4                                     :Value of N
(-1.34, 2.55) ( 0.28, 3.17) (-6.39,-2.20) ( 0.72,-0.92)
(-0.17,-1.41) ( 3.31,-0.15) (-0.15, 1.34) ( 1.29, 1.38)
(-3.29,-2.39) (-1.91, 4.42) (-0.14,-1.35) ( 1.72, 1.35)
( 2.41, 0.39) (-0.56, 1.47) (-0.83,-0.69) (-1.96, 0.67) :End of matrix A
```

## 9.3 Program Results

F07AWF Example Program Results

Inverse

```
                                     1           2           3           4
1 ( 0.0757,-0.4324) ( 1.6512,-3.1342) ( 1.2663, 0.0418) ( 3.8181, 1.1195)
2 (-0.1942, 0.0798) (-1.1900,-0.1426) (-0.2401,-0.5889) (-0.0101,-1.4969)
3 (-0.0957,-0.0491) ( 0.7371,-0.4290) ( 0.3224, 0.0776) ( 0.6887, 0.7891)
4 ( 0.3702,-0.5040) ( 3.7253,-3.1813) ( 1.7014, 0.7267) ( 3.9367, 3.3255)
```

---